Excel for Algebra [1]

Lesson 5: The Solver

OK, what's "The Solver"?

Speaking very informally, the Solver is like "Goal Seek on steroids". It's a lot more powerful, but it's also more challenging to control. (The phrase "'Roid Rage" comes to mind occasionally...)

More formally, the Excel Solver is a general purpose equation solver and optimization package. It can handle linear and nonlinear problems, with multiple unknowns and constraints.

Here are the major differences between Goal Seek and Solver:

Feature	**Goal Seek**	**Solver**
number of unknowns	1	1 to hundreds
number of constraints	1, very limited (set cell to exact value only)	0 to hundreds, very flexible (constrain cell values to be equal, not less than, or not greater than, or restricted to integers or binary choices, with the constraint specified as either a specific value or another cell)
remembers the problem	No – you have to tell it what to do every time.	Yes – the whole problem definition is saved as part of the spreadsheet.
solves optimization problems	No – its only ability is to set a cell to a specified value	Yes – can minimize or maximize one specified "objective function"
safety	safe: will overwrite only numbers	dangerous: will overwrite anything!
adjustable accuracy	Yes, using Options > Calculation > "Maximum Change"	Yes, using options embedded in the Solver dialog.
usability	simple and predictable	more complicated, frequently does something unexpected

Whoa – that's way too much for me! Can we start simple?

Sure! But first...

We need to install the Solver (sigh...)

The Solver has shipped with every copy of Microsoft Excel since 1991. But for some reason, Microsoft chooses not to install it by default. That's annoying, but no big problem — you just have to jump through a couple of extra hoops to make it available.

In Excel 2003, pull down the Tools menu and see if "Solver..." is in it. If it's not, then select Tools > Add-ins, put a check-mark next to Solver, and click OK.

Here's what it looks like:



Now your Tools menu should show that Solver is available.



In Excel 2007, the sequence is to Microsoft Office Button > Excel Options > Add-Ins > Manage Excel Add-ins, click Go, put a check-mark next to Solver, and click OK. The first part looks like this, and the check-marking looks the same as for Excel 2003.



After Solver is installed in Excel 2007, it can be accessed as a button in the Data Ribbon:

OK, now let's get started.

Let's begin with a problem that's just a little bit too complicated for Goal Seek to handle.

Find values for x and y that solve the following system of two equations:

$$0.7x + 0.9y = 2.58$$
$$2.0x + 1.5y = 5.55$$

This sort of problem can be solved by hand, but it's not easy. You have to use a process that's usually called "elimination" or "the addition method". We looked at it briefly in class.

In contrast, it's quite simple to solve with Excel's Solver.

We'll start by setting up one of our usual spreadsheets (shown as formulas, then values):

	A	B	C
1	Variables	x	y
2	Values	2	3
3			
4	Equations	Left Side	Right Side
5			
6	0.7*x + 0.9*y = 2.58	= 0.7*B2 + 0.9*C2	2.58
7	2.0*x + 1.5*y = 5.55	= 2*B2 + 1.5*C2	5.55

	A	B	C
1	Variables	x	y
2	Values	2	3
3			
4	Equations	Left Side	Right Side
5			
6	0.7*x + 0.9*y = 2.58	4.1	2.58
7	2.0*x + 1.5*y = 5.55	8.5	5.55

Notice that we have "pulled out of the air" numbers for x and y. And what a surprise(!), they don't work — the left and right sides of the equations are not equal.

To fix this, we launch Solver by clicking on it in the menu system. That pops up a new dialog called Solver Parameters. By clicking on a few buttons and cells, we make the dialog look like this:



At first glance, this looks complicated. But make your eyes step slowly through the dialog, looking at each field separately, and it gets simpler.

Notice that there's no "Target Cell" – we have to make sure that field is empty.

Then we say to change cells B2 through C2, so that B6 through B7 become equal to C6 through C7, respectively.

In other words, what we've said here is just Excel-speak for "*by changing x and y, make the left side of each equation equal to that same equation's right side*".

When we click the Solve button, the spreadsheet updates to look like this:



Notice that now x=1.5, y=1.7, and the left and right sides of both equations are now equal.

We have solved the problem — no algebra required!

How about another example?

Sure. Let's go back to a mixture problem that we used as an example for Goal Seek. To refresh your memory, the problem was this:

> *Joe Grocer wants to mix 50 pounds of nuts that cost $2 per pound with some other nuts that cost $6 per pound, to make a mixture that costs $5 per pound. How many pounds of $6 nuts should he use?*

Let's change the problem a little bit so that it's natural to think about it in terms of two variables:

> *Joe Grocer wants to mix some nuts that cost $2 per pound with some other nuts that cost $6 per pound, to make 200 pounds of mixture that costs $5 per pound. How many pounds of each kind of nuts must he use?*

The spreadsheet that we'll use for this can actually be quite a bit simpler than what we did with Goal Seek.

To set up for Solver, what we'll do is to simply compute total pounds and total dollars of the mixture by adding the *pounds* and *dollars* of the ingredients. From those, we'll "work backward" to compute the *dollars-per-pound* of the mixture by dividing, instead of multiplying. Then to solve the problem, we'll have Solver adjust the pounds of ingredients as needed to make everything else what it needs to be.

Plucking two numbers from the air for pounds of nuts, our new spreadsheet looks like this:

	A	B	C	D
1	Item	Number of pounds	Dollars per pound	Dollars
2	Type A nuts	20	2	=B2*C2
3	Type B nuts	3	6	=B3*C3
4	mixture	=B3+B2	=D4/B4	=D2+D3

	A	B	C	D
1	Item	Number of pounds	Dollars per pound	Dollars
2	Type A nuts	20	2	40
3	Type B nuts	3	6	18
4	mixture	23	2.521739	58

	A	B	C	D
1	Item	Number of pounds	Dollars per pound	Dollars
2	Type A nuts	20	2	40
3	Type B nuts	3	6	18
4	mixture	23	2.521739	58

Notice that there are no spare calculations in this spreadsheet, and no concept of computing the same number two different ways. We're just computing each of the cells in a direct manner, starting with the number of pounds and dollars per pound of each ingredient.

As usual, it's apparent that the two numbers we plucked from the air do not solve the problem — we have the wrong total weight of mixture (23 pounds versus 200 desired) and also the wrong dollars per pound of mixture ($2.52 per pound versus $5 per pound desired).

So, what we'll do now is have Solver fix up the numbers so that everything becomes the way we want.   What we want is to adjust cells B2 and B3 (pounds of each kind of nuts) so that B4 = 200 (total weight 200 pounds) and C4 = 5 (mixture worth $5 per pound).

There are a couple of different ways that we can do this. One method that is particularly safe and effective is to put the target values for the mixture pounds and dollars-per-pound into the spreadsheet itself.

	A	B	C	D
1	Item	Number of pounds	Dollars per pound	Dollars
2	Type A nuts	20	2	=B2*C2
3	Type B nuts	3	6	=B3*C3
4	mixture	=B3+B2	=D4/B4	=D2+D3
5				
6	target values for mixture	200	5	

Now all we have to do is tell Solver to make the computed values equal to the target values:



Click “Solve”, and the screen updates to say this:



There are other approaches to specifying constraints. You can, for example, put target values in the Solver dialog instead of putting them in the spreadsheet.[2] However, that turns out to be a bad idea.

The method shown here — using no Target Cell and putting all target values in cells of the spreadsheet — has the advantage that it keeps all the numbers visible on the spreadsheet. Also it is <u>much</u> easier to package so that somebody else can use your spreadsheet to solve similar problems with different numbers. You don’t have to explain to them any details about how to set up the Solver, you just tell them where to put target values in the spreadsheet and how to run the Solver. That last part (running the Solver) can even be hidden behind a button if you’re clever about programming Excel macros.

Exercises:

1. Reproduce the examples shown above.

2. Use the Solver to work two other problems from recent homework.

[2] Actually, it’s hard to find documentation that says you don’t need a target cell. Most descriptions of the solver show a target cell being specified, with the target value specified in the Solver parameters. But if you look hard enough, you can find “Design and Use of the Microsoft Excel Solver” written by the software authors themselves (for example at http://www.utexas.edu/courses/lasdon/design3.htm). It explains that if you specify Set Target Cell To Value, all that really happens is that the solver adds an equality constraint. In other words, there’s no difference to the guts of the Solver where you specify the equality — that difference is only in the user interface.

A Challenging Example of How to Use Solver: Building a nutritionally balanced avocado-burger

Now let's consider a really challenging problem.

> One specific diet calls for 50% calories from carbohydrates, 30% calories from fat, and 20% calories from protein. Suppose you want to make an avocado-burger that exactly meets the recommendations. An avocado-burger consists of exactly one full bun, smeared with some amount of avocado and a (possibly very small) hamburger patty. **How much hamburger and avocado do you put with each bun?**
>
> Use the following nutritional analysis of buns, avocado, and hamburger. You can assume that all of the remaining calories come from carbohydrate.
>
> Buns (package label for Franz "9 Whole Grain")
> Serving size 85 grams
> Total calories 220
> Calories from fat 30
> Calories from protein 45
>
> Avocado (nutritiondata.com for "California avocado")
> Serving size 230 gm
> Total calories 384
> Calories from fat 297
> Calories from protein 25
>
> Hamburger (nutritiondata.com for 90% lean broiled)
> Serving size 28gm
> Total calories 61
> Calories from fat 30
> Calories from protein 31

One good way to solve this problem is to set up a spreadsheet that

1. computes the calorie contributions of each ingredient, by scaling in proportion to the amount used (example: avocado fat calories = $GramsUsed \times \frac{297\ calories}{230\ gm}$),
2. adds up the calorie contributions of all three ingredients to get the total calories in the sandwich,
3. divides each of calorie sums (carbohydrate, fat, protein) by the total calories, to get percent calories of each type, and
4. uses Solver to find the amounts used, that make the percent calories be the requested values.

Here is the spreadsheet (with solved values shown):

	A	B	C	D	E	F	G	H	I	J	K
1	**Planning a nutrient-balanced avocado-burger**										
2											
3											
4		**Nutritional analysis of ingredients (from tables)**						**Analysis, scaled to amount used**			
5	**Ingredients**	**Specified size (gm)**	**Total Calories in specified size**	**Calories from carbs in specified size**	**Calories from fat in specified size**	**Calories from protein in specified size**	**Amount used (gm) -- Adjust These To Fit**	**Total calories in amount used**	**Calories from carbs in amount used**	**Calories from fat in amount used**	**Calories from protein in amount used**
6											
7	**bun**	85	220	145	30	45	85	220	145	30	45
8	**avocado**	230	384	62	297	25	**39.3537**	65.70357	10.60839	50.8176	4.277576
9	**90% lean hamburger**	28	61	0	30	31	**11.711**	25.51321	0	12.54748	12.96573
10											
11											
12				**Total in sandwich (computed)**				311.2168	155.6084	93.36508	62.2433
13				**% in sandwich (computed)**					50.0%	30.0%	20.0%
14				**% in sandwich (TARGET)**					50%	30%	20%

	A	B	C	D	E	F	G	H	I	J	K
1	**Planning a nutrient-ba**										
2											
3											
4		**Nutritional analysis of ingredients (from tables)**						**Analysis, scaled to amount used**			
5	**Ingredients**	**Specified size (gm)**	**Total Calories in specified size**	**Calories from carbs in specified size**	**Calories from fat in specified size**	**Calories from protein in specified size**	**Amount used (gm) -- Adjust These To Fit**	**Total calories in amount used**	**Calories from carbs in amount used**	**Calories from fat in amount used**	**Calories from protein in amount used**
6											
7	**bun**	85	220	=C7-E7-F7	30	45	85	=$G7/$B7*C7	=$G7/$B7*D7	=$G7/$B7*E7	=$G7/$B7*F7
8	**avocado**	230	384	=C8-E8-F8	297	25	**39.353700**	=$G8/$B8*C8	=$G8/$B8*D8	=$G8/$B8*E8	=$G8/$B8*F8
9	**90% lean hamburger**	28	61	=C9-E9-F9	30	31	**11.710980**	=$G9/$B9*C9	=$G9/$B9*D9	=$G9/$B9*E9	=$G9/$B9*F9
10											
11											
12				**Total in san**				=SUM(H7:H9)	=SUM(I7:I9)	=SUM(J7:J9)	=SUM(K7:K9)
13				**% in sandw**					=I12/$H$12	=J12/$H$12	=K12/$H$12
14				**% in sandw**					0.5	0.3	0.2

The solver specification in this case consists of:



As with our simpler mixture of nuts, we have specified the entire problem as cells to change (G8:G9, the amounts of avocado and hamburger that we were asked to determine) and Constraints (I13:K13 = I14:K14, meaning computed percentages = target percentages).

Exercises:

- Change the target percentages to 30%, 50%, 20% and repeat the solution:

4		Nutritional analysis of ingredients (from tables)						Analysis, scaled to amount used			
5	Ingredients	Specified size (gm)	Total Calories in specified size	Calories from carbs in specified size	Calories from fat in specifie d size	Calories from protein in specified size	Amount used (gm) -- Adjust These To Fit	Total calories in amount used	Calories from carbs in amount used	Calories from fat in amount used	Calories from protein in amount used
6											
7	bun	85	220	145	30	45	85	220	145	30	45
8	avocado	230	384	62	297	25	177.509	296.3635	47.85035	229.2186	19.2945
9	90% lean hamburger	28	61	0	30	31	58.0526	126.4718	0	62.19925	64.27256
10											
11											
12				Total in sandwich (computed)				642.8353	192.8504	321.4179	128.5671
13				% in sandwich (computed)					30.0%	50.0%	20.0%
14				% in sandwich (TARGET)					30%	50%	20%

Notice that by allowing yourself to have a higher percentage of fat, you get to have a <u>lot</u> more avocado and hamburger to go with your bun.

- Change the target percentages to 70%, 15%, 15% and repeat the solution. Solver will not complain, but you'll get some strange numbers:

4		Nutritional analysis of ingredients (from tables)						Analysis, scaled to amount used			
5	Ingredients	Specified size (gm)	Total Calories in specified size	Calories from carbs in specified size	Calories from fat in specifie d size	Calories from protein in specified size	Amount used (gm) -- Adjust These To Fit	Total calories in amount used	Calories from carbs in amount used	Calories from fat in amount used	Calories from protein in amount used
6											
7	bun	85	220	145	30	45	85	220	145	30	45
8	avocado	230	384	62	297	25	**12.2868**	20.51355	3.312084	15.86595	1.335518
9	90% lean hamburger	28	61	0	30	31	**-13.146**	-28.63925	0	-14.0849	-14.5544
10											
11											
12				Total in sandwich (computed)				211.8743	148.3121	31.78107	31.78115
13				% in sandwich (computed)					70.0%	15.0%	15.0%
14				% in sandwich (TARGET)					70%	15%	15%

What's happening here is that you've asked for a higher percentage of carbohydrates in the whole sandwich than there is in any of the ingredients! Mathematically, Solver can meet this requirement by using a negative amount of hamburger. Of course you can't actually have a negative amount of hamburger, so this is a non-physical solution. (We could have kept Solver from generating such a "solution" by adding constraints for ">=0" on the amounts. In that case it would have said that it couldn't find a feasible solution.)

- Change the target percentages to 70%, 30%, 30% (yep, two 30's), and repeat the solution. This time, you should get a response that "Solver could not find a feasible solution". Well, yeah. We set up the spreadsheet so that the % calories have to sum to 100%. Telling Solver to find a solution where the sum is 130% isn't going to work!